Maxima y GeoGebra para problemas de Selectividad

Tema 4. Resolución de problemas con ordenador

Matemáticas en el Paradigma Educativo Actual
Máster de Formación del Profesorado, UNED

Antonio García Sevilla

Septiembre 2019

1. Introducción

Hoy en día, gracias al uso extendido de los ordenadores y de internet, disponemos de multitud de herramientas computacionales, muchas de ellas de libre acceso, que permiten realizar cálculos y algoritmos matemáticos de manera eficiente, correcta, y muy rápidamente. Estas herramientas son a menudo de utilidad para los matemáticos profesionales, permitiéndoles acelerar muchas tareas, o asistiendo con las partes más rutinarias.

En la enseñanza de las Matemáticas, estas herramientas también pueden resultar muy útiles. Tanto para asistir al profesor en el planteamiento de problemas y ejercicios, como durante la explicación en clase, como para servir de apoyo al alumno en casa, cuando no tiene al profesor al lado para corregir los cálculos o proveer intuiciones que ayuden al estudio.

En este documento, usamos dos herramientas disponibles de manera gratuita, Maxima y GeoGebra, como ejemplo del uso de este tipo de programas en la educación secundaria. Usamos cada uno con un problema del examen de Matemáticas II de la EvAU en la Comunidad de Madrid para el curso 2018-2019, convocatoria ordinaria[3], mostrando cómo pueden ser útiles para alumnos de 2^o de Bachillerato, aunque su uso se puede extender a otros cursos de la secundaria.

2. Cálculo matricial con Maxima

Usaremos Maxima[2] para resolver el Ejercicio 1 de la opción A. El enunciado es:

Dadas la matrices $A = \begin{pmatrix} 1 & 3 & 4 & 1 \\ 1 & a & 2 & 2-a \\ -1 & 2 & a & a-2 \end{pmatrix}$ y $M = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 1 \end{pmatrix}$, se pide:

a) (1.5 puntos) Estudiar el rango de A en función del parámetro real a.

b) (1 punto) Calcular, si es posible, la inversa de la matriz AM para el caso $a = 0$.

Puesto que es un ejercicio de estudio, Maxima no lo puede resolver por completo por nosotros. Pero puede ayudarnos a hacer el cálculo mecánico, lo que nos ahorra tiempo y nos permite enfocarnos más al estudio analítico, que requiere de pensar y entender, los verdaderos valores que este ejercicio pretende comprobar, más que la habilidad algebraica para resolver determinantes.

La solución del ejercicio usando Maxima está en el fichero `matrices.mac`, ejecutable directamente por Maxima en modo "batch", y lo incluimos a continuación:

```
/* Definimos las matrices A y M del enunciado. En Maxima,
 * escribimos el nombre seguido de dos puntos para definir
 * algo. Después usamos el comando 'matrix', y le pasamos
 * las filas entre paréntesis, separadas por comas. Cada
 * fila es un 'vector', que se escribe entre corchetes, los
 * elementos también separados por comas.
 */
A: matrix ([1, 3, 4, 1], [1, a, 2, 2-a], [-1, 2, a, a-2]);
M: matrix ([1, 0, 0], [0, 1, 0], [0, 0, 0], [0, 0, 1]);

/* Podemos explorar "cómo se ve" la matriz A, dependiendo
 * del parámetro a, evaluándolo en distintos puntos. Esto
 * nos puede dar una idea intuitiva de cómo van a ser los
 * cálculos, y con Maxima es muy fácil. Evaluamos A en
 * valores de a que nos parecen interesantes: 2 y 0.
 */
A, a=2;
A, a=0;

/* Y podemos hacer lo mismo con el rango: */
rank(A), a=2;
rank(A), a=0;
```

```
/* En general, como la matriz A es 3x4, el rango máximo es
 * 3. Además, es una matriz no nula, por lo que al menos
 * será uno. Y hay un subdeterminante no nulo de orden 2 muy
 * fácil de ver:
 */
submatrix(3, A, 2, 4);
/* Le hemos pedido a Maxima que ignore la fila 3, y las
 * columnas 2 y 4 (al revés de cómo lo podríamos pensar
 * nosotros, que es coger las filas 1 y 2 y columnas 1 y 3,
 * pero es equivalente. Luego calculamos el determinante, y
 * como es distinto de 0 (-2), y no depende de a, el rango
 * de A será al menos 2 en todo caso.
 */
determinant(%);

/* Por tanto, lo que nos queda es buscar menores de orden 3
 * y estudiarlos para los valores de a. Elegimos por
 * comodidad las tres primeras columnas (para Maxima, esto
 * es quitar la columna 4):
 */
submatrix(A, 4);
/* Calculamos el determinante, y hallamos las raíces
 * resolviendo el polinomio para la variable a:
 */
determinant(%);
solve(%, a);

/* Maxima nos dice que hay dos valores donde se anula, a=1 y
 * a=-2. Es decir, para todo otro valor, este determinante
 * será no nulo y el rango de A será 3.  Ahora podemos
 * calcular el rango de A directamente con Maxima para esos
 * valores de a, puesto que ya no es una incógnita:
 */
rank(A), a=1;
rank(A), a=-2;

/* Por tanto, la solución al primer apartado es que el rango
 * de A será 3 para todo a distinto de 1 y -2, y 2 para esos
 * valores.
 */

/* Para el segundo apartado, primero obtenemos el producto
 * de A y M. En Maxima, la multiplicación de matrices se
 * representa con un punto, es decir:
```

```
 */
A.M;

/* Como nos dicen el valor de a, se lo pasamos también a
 * Maxima, y le damos el nombre 'B' */
B: A.M, a=0;

/* Calculamos el determinante para comprobar que es
 * invertible: */
determinant(B);

/* Y finalmente hacemos el cálculo que nos piden: */
invert(B);
```

3. Rectas y Planos en el espacio con GeoGebra

En esta sección, usaremos GeoGebra[4] para visualizar la resolución del Ejercicio 3 de la opción A. El enunciado es:

> Dadas las recta $r \equiv \frac{x-1}{2} = \frac{y-3}{-2} = z$, y la recta s que pasa por el punto $(2,-5,1)$ y tiene dirección $(-1,0,-1)$, se pide:
>
> a) (1 punto) Estudiar la posición relativa de las dos rectas.
>
> b) (1 punto) Calcular un plano que sea paralelo a r y que contenga a s.
>
> c) (0.5 puntos) Calcular un plano perpendicular a la recta r y que pase por el origen de coordenadas.

En este caso, nuestro interés no es tanto resolver el ejercicio de manera algebraica, sino visualizar en el espacio las relaciones entre los elementos implicados. Esto nos ayuda a entender realmente el porqué de los distintos cálculos, sin perdernos en el ejemplo concreto, que puede ser más o menos complicado.

Ver lo que estamos realizando, en directo y en el espacio, nos permitirá crear el entendimiento geométrico que hará que sepamos qué hacer ante un problema futuro, y hacer los cálculos con la seguridad de entender que está ocurriendo.

Primero abrimos la aplicación, en nuestro caso la versión *online*[1], y seleccionamos "Graficadora 3D" en el menú superior derecho. Esto nos presenta un espacio de trabajo tridimensional, con los ejes dibujados y el plano $z=0$ como referencia:

[1] `https://www.geogebra.org/`



Podemos usar el ratón para desplazar y rotar el espacio, lo que nos permite ver la escena desde distintos ángulos, y observar las posiciones relativas de los distintos objetos. A la izquierda tenemos la “Calculadora”, donde vamos a introducir todos los comandos.

Empezamos creando los objetos del enunciado. Pulsamos en la primera celda de la calculadora, y escribimos:

```
r: (x-1)/2 = (y-3)/-2 = z
```

Con los dos puntos creamos el objeto y le damos nombre, y la calculadora automáticamente va colocando la ecuación en un formato adecuado. También vemos al acabar de escribir, que la recta se ha dibujado en el espacio, y debajo de nuestra ecuación tenemos la forma paramétrica equivalente de r.



Para añadir s, simplemente escribimos la forma paramétrica con el punto y dirección que nos dan:

```
s: (2,-5,1) + t * (-1, 0, -1)
```

Usamos t como parámetro por ser fácil de escribir, pero vemos que GeoGebra lo reescribe automáticamente como λ.

Ahora que tenemos las dos rectas, podemos ver claramente que no se cortan, sobre todo si rotamos la escena con el ratón y vemos el punto de máxima aproximación. Al cambiar el punto de vista, podemos ver bajo distintas perspectivas la relación de las rectas, y entender el resultado algebraico que habríamos obtenido que nos dice que las rectas no son paralelas ni se cortan, sino que se cruzan.





Esto nos soluciona el apartado a, así que pasamos al b. Para calcular un plano paralelo a r y que contenga a s con GeoGebra, podemos usar la herramienta visual “Plano que pasa por tres puntos” (haciendo click en el icono a la derecha de la calculadora), o la función *Plano*.

Si elegimos dos puntos sobre s, significará que el plano (lo llamaremos P) contendrá automáticamente a esa recta, por lo que cogeremos A y B de esa manera[2]. Para ello, puedes utilizar la herramienta visual “Punto”, o escribir los siguientes comandos:

```
A = Punto(s, 0.5)
B = Punto(s, 0.2)
```

Estos comandos eligen dos puntos sobre s. El parámetro que le damos a GeoGebra no es el de la fórmula paramétrica, sino uno interno del programa, pero nos vale cualquiera que nos dé dos puntos distintos.

El tercer punto lo tenemos que elegir con cuidado, para que el plano resultante sea paralelo a r. Para ello, cogemos el punto A y lo desplazamos según el vector director de r, que GeoGebra ya ha calculado y tenemos escrito debajo de la definición de r:

[2]Si tienes dudas sobre esto, compruébalo en GeoGebra haciendo distintos planos con un tercer punto.

```
C = A + (-0.5, 0.5, -0.25)
```



Ahora ya podemos calcular el plano que nos piden, dándole a GeoGebra los tres puntos a usar, y girando la escena podemos comprobar que, efectivamente, contiene a s y es paralelo a r:

```
P: Plano(A, B, C)
```



Solucionado el apartado b, pasamos al c, que podemos resolver directamente con una función de GeoGebra:

```
Q: PlanoPerpendicular((0, 0, 0), r)
```





La calculadora nos ha quedado al final con todos los objetos que hemos ido creando, así como las formas que ha calculado GeoGebra, que podemos usar para comprobar que nuestros cálculos han sido correctos.



Obviamente, el ejercicio que hemos hecho no es sustituto a hacer los cálculos, entre otras cosas porque en el examen no tendremos el ordenador disponible. Sin embargo, es una herramienta muy útil para visualizar lo que estamos haciendo, de manera muy sencilla e inmediata, por lo que nos puede ayudar a entender cuando en algún ejercicio obtengamos resultados inesperados, o no sepamos por dónde empezar.

El ejercicio resuelto está disponible en el fichero `planosyrectas.ggb`, y se puede abrir en GeoGebra con el botón Abrir (icono de lupa).

4. Conclusión

Como hemos visto, el uso de dos programas matemáticos gratuitos y de libre acceso puede suponer un gran apoyo a la hora de resolver problemas y ejercicios. En el caso de Maxima, realiza los cálculos algebraicos de manera intuitiva y muy rápida, permitiéndonos comprobar que el proceso a seguir es el correcto, y verificando los resultados obtenidos a mano.

GeoGebra es especialmente potente por la interactividad visual que aporta. Muchos conceptos geométricos se entienden mucho mejor visualmente que algebraicamente, y este apoyo interactivo a la resolución de problemas de geometría, que en el espacio pueden ser complicados de dibujar, es insuperable.

En cuanto a si alguno es mejor que el otro, la conclusión es claramente que cada uno tiene sus puntos fuertes, y brillan en distintos tipos de problemas, por lo que lo ideal es usar cada uno cuando sea oportuno. La velocidad de cálculo y potencia simbólica de Maxima es perfecta para problemas de matrices, donde los cálculos pueden ser muy tediosos. Y aunque Maxima también nos permite resolver las cuestiones de geometría algebraicamente, y visualizar los resultados, la aproximación "táctil" de GeoGebra es probablemente mejor para estimular la visión espacial y la intuición geométrica.

Bibliografía

[1] Richard H. Rand (2005): Introduction to Maxima: Matrix calculations. Disponible en `http://eagle.cs.kent.edu/MAXIMA/maxima_27.html`, consultado el 20 de Agosto de 2019.

[2] Jaime Villate (2018): Maxima 5.42.0 Manual. Disponible en `http://maxima.sourceforge.net/docs/manual/maxima.html`, consultado el 20 de Agosto de 2019.

[3] Universidades Públicas de la Comunidad de Madrid (2019): Evaluación para el Acceso a las Enseñanzas Universitarias Oficiales de Grado. Materia: Matemáticas II. Curso 2018-2019,convocatoria ordinaria. Disponible en `https://www.ucm.es/data/cont/docs/3-2019-06-05-2MATEM%C3%81TICAS%20II.pdf`.

[4] International GeoGebra Institute (2019): Manual de GeoGebra. Disponible en `https://wiki.geogebra.org/es/Manual`, consultado el 20 de Agosto de 2019.